

**Concordia University
Department of Computer Science
and Software Engineering**

**Compiler Design (COMP 442/6421)
Winter 2023**

Assignment 2, Syntactic Analyzer

Deadline:	Sunday February 19 th , 2023
Evaluation:	8% of final grade
Late submission:	penalty of 50% for each late working day

This assignment is about the design and implementation of a syntactic analyzer for the language specified by the grammar given below. The syntactic analyzer should use as input the token stream produced by the lexical analyzer that you have produced in assignment #1, and prove whether or not the token stream is a valid program according to the grammar given below. While doing so, it should locate, report, and recover from eventual syntax errors. It should also write to a file a trace of the derivation that is proving that the input token stream can be derived from the starting symbol of the grammar.

The assignment includes two grading source files. These files should be used as-is and not be altered in any way. Completeness of testing is a major grading topic. You are responsible for providing appropriate test cases that test for a wide variety of valid and invalid cases in addition to what is in the grading source files provided.

Grammar

$G = (N, T, S, R)$

N – Nonterminal Symbols

addOp, aParams, aParamsTail, arithExpr, arraySize, assignOp, assignStat, classDecl, classDeclOrFuncDef, expr, factor, fParams, fParamsTail, funcBody, funcDef, funcHead, functionCall, idnest, indice, localVarDecl, localVarDeclOrStmt, memberDecl, memberFuncDecl, multOp, relExpr, relOp, returnType, sign, START, statBlock, statement, term, type, variable, visibility

T – Terminal Symbols

,, +, -, or, [, intLit,], =, class, id, {, }, ;, (,), floatLit, not, :, void, ., *, /, and, isa, eq, geq, gt, leq, lt, neq, if, then, else, read, return, while, write, float, integer, private, public, function, arrow, constructor, attribute, sr, localVar

S – Starting Symbol

START

R – Rules

```

<START> ::= {{<classDeclOrFuncDef>}}
<classDeclOrFuncDef> ::= <classDecl>
                        | <funcDef>
<classDecl> ::= 'class' 'id' [[ 'isa' 'id' {{',' 'id'}} ]] '{' {{<visibility> <memberDecl>}} '}' ';';
<visibility> ::= 'public' | 'private' | EPSILON
<memberDecl> ::= <memberFuncDecl>
                | <memberVarDecl>
<memberFuncDecl> ::= 'function' 'id' ':' '(' <fParams> ')' 'arrow' <returnType> ';';
                | 'constructor' ':' '(' <fParams> ')' ';';
<memberVarDecl> ::= 'attribute' 'id' ':' <type> {{<arraySize>}} ';';
<funcDef> ::= <funcHead> <funcBody>
<funcHead> ::= 'function' [[ 'id' 'sr' ]] 'id' '(' <fParams> ')' 'arrow' <returnType>
                | 'function' 'id' 'sr' 'constructor' '(' <fParams> ')'
<funcBody> ::= '{' {{<localVarDeclOrStmt>}} '}'
<localVarDeclOrStmt> ::= <localVarDecl>
                        | <statement>
<localVarDecl> ::= 'localVar' 'id' ':' <type> {{<arraySize>}} ';';
                | 'localVar' 'id' ':' <type> '(' <aParams> ')' ';';
<statement> ::= <assignStat> ';';
                | 'if' '(' <relExpr> ')' 'then' <statBlock> 'else' <statBlock> ';';
                | 'while' '(' <relExpr> ')' <statBlock> ';';
                | 'read' '(' <variable> ')' ';';
                | 'write' '(' <expr> ')' ';';
                | 'return' '(' <expr> ')' ';';
                | <functionCall> ';';
<assignStat> ::= <variable> <assignOp> <expr>
<statBlock> ::= '{' {{<statement>}} '}' | <statement> | EPSILON
<expr> ::= <arithExpr> | <relExpr>
<relExpr> ::= <arithExpr> <relOp> <arithExpr>
<arithExpr> ::= <arithExpr> <addOp> <term> | <term>
<sign> ::= '+' | '-'
<term> ::= <term> <multOp> <factor> | <factor>
<factor> ::= <variable>
                | <functionCall>
                | 'intLit' | 'floatLit'
                | '(' <arithExpr> ')'
                | 'not' <factor>
                | <sign> <factor>
<variable> ::= {{<idnest>}} 'id' {{<indice>}}
<functionCall> ::= {{<idnest>}} 'id' '(' <aParams> ')'
<idnest> ::= 'id' {{<indice>}} '.'
                | 'id' '(' <aParams> ')' '.'
<indice> ::= '[' <arithExpr> ']'
<arraySize> ::= '[' 'intLit' ']' | '[' ']'
<type> ::= 'integer' | 'float' | 'id'
<returnType> ::= <type>
                | 'void'
<fParams> ::= 'id' ':' <type> {{<arraySize>}} {{<fParamsTail>}} | EPSILON
<aParams> ::= <expr> {{<aParamsTail>}} | EPSILON
<fParamsTail> ::= ',' 'id' ':' <type> {{<arraySize>}}
<aParamsTail> ::= ',' <expr>
<assignOp> ::= '='
<relOp> ::= 'eq' | 'neq' | 'lt' | 'gt' | 'leq' | 'geq'
<addOp> ::= '+' | '-' | 'or'
<multOp> ::= '*' | '/' | 'and'

```

Notes

- Terminals (lexical elements, or tokens) are represented in single quotes *'LikeThis'*.
- Non-terminals are represented between angle brackets *<LikeThis>*.
- The empty phrase is represented by *EPSILON*.
- EBNF-style repetition notation is represented using double curly brackets *{{Like this}}*. It represents zero or more occurrence of the sentential form enclosed in the brackets.
- EBNF-style optionality notation is represented using double square brackets *[[Like this]]*. It represents zero or one occurrence of the sentential form enclosed in the brackets.
- *id* follows the specification for program identifiers found in assignment #1.
- *intLit*, *floatLit*, follow specification for integer and float literals found in assignment #1

Work to submit

Document

You must provide a short document that includes the following sections:

- Section 1. Transformed grammar into LL(1)** : Remove all the EBNF notations and replace them by right-recursive list-generating productions. Analyze the syntactical definition (using tools) and list in your documentation all the ambiguities and left recursions. Modify the grammar so that the left recursions and ambiguities are removed without modifying the language. Include in your documentation the set of productions that can be parsed using the top-down predictive parsing method, i.e. an LL(1) grammar.
- Section 2. FIRST and FOLLOW sets** : Derive the FIRST and FOLLOW sets for each non-terminal in your transformed grammar.
- Section 3. Design** : Give a brief overview of the overall structure of your solution, as well as a brief description of the role of each component of your implementation.
- Section 4. Use of tools** : Identify all the tools/libraries/techniques that you have used in your analysis or implementation and justify why you have used these particular ones as opposed to others.

Implementation

- **Parser** : Implement a predictive parser (recursive descent or table-driven) for your modified set of grammar rules.
- The result of the parser should be the creation of a derivation proving that the grammar can generate the program given in input. When parsing a file named, for example, **originalfilename**, the parser should write into a file named **originalfilename.outderivation** the derivation proving that the grammar can generate the original program. When syntax errors are found, error messages should be output in a file named **originalfilename.outsyntaxerrors**.
- **Derivation output** : Your parser should write to a file the leftmost derivation that proves that the source program can be derived from the starting symbol. For an example of a derivation, see slide set 3, slide 17.
- **Error reporting** : The parser should properly identify all the errors in the input program and print a meaningful message to the user for each error encountered. The error messages should be informative on the nature of the errors, as well as the location of the errors in the input file.
- **Error recovery** : The parser should implement an error recovery method that permits to report all errors present in the source code.
- **Test cases** : Write a set of source files that enable to test the parser for all syntactical structures involved in the language. Include cases testing for a variety of different errors to demonstrate the accuracy of your error reporting and recovery.
- **Driver**: Include a driver that parses all your test files. For each test file, the corresponding **outsyntaxerrors**, and **outderivation** files should be generated.

Assignment submission requirements and procedure

- Each submitted assignment should contain four components: (1) the source code, (2) a group of test files, (3) a brief report, and (4) an executable named **parserdriver**, that extracts the tokens from all your test files. For each test file, the corresponding **outsyntaxerrors** and **outderivation** files should be generated.
- The assignment statement provides test files (**.src**) and their corresponding output files.
- The source code should be separated into modules using a comprehensible coding style.
- The assignment should be submitted through moodle in a file named: "**A#_student-id**" (e.g. **A1_1234567**, for Assignment #1, student ID 1234567) and the report must in a PDF format.
- You may use any language you want in the project and assignments but the only fully supported language during the lab is Java.
- You have to submit your assignment before midnight on the due date on moodle.
- The file submitted must be a **.zip** file

The marking will be done in a short presentation to the marker. A schedule will be provided to you by email in the days before the due date. You will be given a short time for the presentation, so make sure that you are ready to effectively demonstrate all the elements mentioned in the "Work to submit" section above.

Evaluation criteria and grading scheme

Analysis:		
List of left recursions and ambiguities in the original grammar, and then resulting correctly transformed LL(1) grammar – document Section 1.	ind 2.1	3 pts
FIRST and FOLLOW sets of the transformed grammar – document Section 2.	ind 2.2	2 pts
Design/implementation:		
Description/rationale of the overall structure of the solution and the roles of the individual components used in the applied solution – document Section 3.	ind 4.3	2 pts
Correct implementation of a top-down predictive parser following the grammar given in this handout.	ind 4.4	15 pts
Output of clear error messages (error description and location) in an outsyntaxerrors file.	ind 4.4	3 pts
Output of a derivation in an outderivation file.	ind 4.4	5 pts
Implementation of an error recovery mechanism.	ind 4.4	3 pts
Completeness of test cases.	ind 4.4	12 pts
Use of tools:		
Description of tools/libraries/techniques used in the analysis/implementation. Description of other tools that might have been used. Justification of why the chosen tools were selected – document Section 4.	ind 5.2	2 pts
Successful/correct use of tools/libraries/techniques used in the analysis/implementation.	ind 5.1	3 pts
Total		50 pts